Multivariate Models for Repeated Measures Response Times in Older and Younger Adults

These data were collected as part of my masters' thesis, and are unpublished in this form (to see the way I'd prefer to have analyzed the data, see Hoffman & Rovine, 2007 *Behavior Research Methods*). The outcome was the log-transformed mean per condition of response time to detect changes in driving scenes that were either of low/high meaningfulness to driving or low/high visual salience (i.e., a 2x2 repeated measures design). This sample includes 97 younger adults (age range= 18–32) and 59 older adults (age range = 63–86). We will specify piecewise effects of age that create mean differences between younger and older adults as well as the effect of age within the older adults.

Original data in <u>multivariate</u> format (was one row per person, outcomes in separate columns):

	PersonID: Person ID number	old: Is in Older Age Group 0=YA, 1=OA)	age: Actual Age in Years	rt11: Response Time (sec) for Low Meaning, Low Salience	rt12: Response Time (sec) for Low Meaning, High Salience	rt21: Response Time (sec) for High Meaning, Low Salience	rt22: Response Time (sec) for High Meaning, High Salience
97	112	0	27.00	12.410	5.524	10.114	7.435
98	201	1	77.00	15.087	10.099	15.957	13.502

New data in <u>stacked</u> format (one row per outcome per person) after transformation code below:

	PersonID: Person ID number	old: Is in Older Age Group 0=YA, 1=OA)	age: Actual Age in Years	condition: Index for Outcome (1-4)	Meaning (1=Low, 2=High)	Salience (1=Low, 2=High)	rt: Combined Response Time across Conditions	logRT: Natural Log of Response Time	yrs65: Age in Older Adult Group (0=65)
385	112	0	27.00	1	1Low	1Low	12.410333333	2.5185294589	0
386	112	0	27.00	2	1Low	1Low	5.5239583333	1.7090946927	0
387	112	0	27.00	3	2High	1Low	10.113680556	2.3138890178	0
388	112	0	27.00	4	2High	2High	7.435	2.0061985799	0
389	201	1	77.00	1	1Low	1Low	15.086736111	2.7138159546	12
390	201	1	77.00	2	1Low	1Low	10.098571429	2.3123939711	12
391	201	1	77.00	3	2High	1Low	15.956517857	2.7698673888	12
392	201	1	77.00	4	2High	2High	13.502083333	2.6028439945	12

SPSS Syntax for Stacking into Univariate (now one row <u>per outcome per person</u>):

```
* Define location of files used in code below.
FILE HANDLE filesave /NAME = "C:\Dropbox\14_SPLH861\861_Example5".
* Import example 5 multivariate data into work library and stack it.
GET FILE = "filesave/SPSS_Example5.sav".
DATASET NAME Example5 WINDOW=FRONT.
VARSTOCASES
   /MAKE rt FROM rt11 rt12 rt21 rt22
   /INDEX = condition (4)
   /KEEP = ALL.
* Create condition variables.
DO IF (condition=1).
COMPUTE mean=1.
COMPUTE sal=1.
END IF.
DO IF (condition=2).
COMPUTE mean=1.
COMPUTE sal=2.
END IF.
DO IF (condition=3).
COMPUTE mean=2.
```

```
COMPUTE sal=1.
END IF.
DO IF (condition=4).
COMPUTE mean=2.
COMPUTE sal=2.
END IF.
* Label new stacked variables.
VARIABLE LABELS
condition "condition: Index for Outcome (1-4)"
mean "Meaning (1=Low, 2=High)"
sal "Salience (1=Low, 2=High)"
rt "rt: Combined Response Time across Conditions".
* Create value labels for conditon variables.
VALUE LABELS mean sal 1 "1Low" 2 "2High".
 * Create variables for analysis.
COMPUTE logrt=LN(rt).
IF (old=0) yrs65=0.
IF (old=1) yrs65=age-65.
* Label new variables.
VARIABLE LABELS
logrt "logRT: Natural Log of Response Time"
yrs65 "yrs65: Age in Older Adult Group (0=65)".
EXECUTE.
```

STATA Syntax for Stacking into Univariate (now one row <u>per outcome per person</u>):

```
* Defining global variable for file location to be replaced in code below
global filesave "C:\Dropbox\14_SPLH861\861_Example5"
* Import example 5 multivariate data into work library and stack it
* List multivariate variables first, i(personID) j(condition)
use "$filesave\STATA_Example5.dta", clear
reshape long rt, i(personid) j(condition)
* Create condition variables
gen mean=1
gen sal=1
recode mean (1=2) if condition==21
recode mean (1=2) if condition==22
recode sal  (1=2) if condition==12
recode sal  (1=2) if condition==22
* Label new stacked variables
label variable condition "condition: Index for Outcome"
label variable mean "Meaning (1=Low, 2=High)"
label variable sal "Salience (1=Low, 2=High)"
label variable rt "rt: Combined Response Time across Conditions"
* Create value labels for condition variables
label define fcondition 1 "1Low" 2 "2High"
label values mean sal fcondition
* Create variables for analysis
gen logrt=ln(rt)
gen yrs65=0
replace yrs65=age-65 if old==1
* Label new variables
label variable logrt "logRT: Natural Log of Response Time"
label variable yrs65 "yrs65: Age in Older Adult Group (0=65)"
```

SAS Syntax for Stacking into Univariate (now one row <u>per outcome per person</u>):

```
* Import example 5 multivariate data into work library and stack it;
DATA work.Example5; SET filesave.SAS_Example5;
       condition=1; mean=1; sal=1; rt=rt11; OUTPUT; * Low  meaning, low  salience;
       condition=2; mean=1; sal=2; rt=rt12; OUTPUT; * Low  meaning, high salience;
       condition=3; mean=2; sal=1; rt=rt21; OUTPUT; * High meaning, low  salience;
       condition=4; mean=2; sal=2; rt=rt22; OUTPUT; * High meaning, high salience;
* Label new stacked variables;
```

```
LABEL condition= "condition: Index for Outcome (1-4)"
      mean= "Meaning (1=Low, 2=High)"
      sal= "Salience (1=Low, 2=High)"
      rt= "rt: Combined Response Time across Conditions";
* Drop old multivariate outcomes;
       DROP rt11--rt22;
RUN;
* Create format (like value label) to use for condition variables;
PROC FORMAT; VALUE fcondition 1="1Low" 2="2High"; RUN;
* Create variables for analysis;
DATA work.Example5; SET work.Example5;
* Log RT to improve residual normality;
  logRT=LOG(RT);
* Format condition variables;
  FORMAT mean sal fcondition.;
* Create piecewise slope for age;
       IF old=0 THEN yrs65=0;
  ELSE IF old=1 THEN yrs65=age-65;
* Label new variables;
  LABEL logrt= "logRT: Natural Log of Response Time"
        yrs65= "yrs65: Age in Older Adult Group (0=65)";
RUN;
```

Empty Multivariate Model Predicting Log RT: This model predicts the RT in condition *c* for person *i* →

$$LogRT_{ci} = \beta_0 + \beta_1 Mean_{ci} + \beta_2 Sal_{ci} + \beta_3 Mean_{ci} * Sal_{ci} + e_{ci}$$

Although this model doesn't look empty, it is—each <u>outcome</u> has its own mean with no other predictors. Condition means are thus created by:

	Low Salience	High Salience
Low Meaning	$\beta_0 + \beta_1 + \beta_2 + \beta_3$	$\beta_0 + \beta_1$
High Meaning	$\beta_0 + \beta_2$	β_0

Let's start with the "answer key" model for the variance: An unstructured R matrix in which all variances and covariances across the four outcomes are estimated separately ("multivariate" ANOVA):

```
ECHO 'SPSS Empty Multivariate Model: RT Mean Differences for Meaning by Salience;'.
ECHO 'Unstructured R Matrix'.
MIXED logrt BY PersonID condition mean sal
     /METHOD   = REML
     /PRINT    = SOLUTION TESTCOV R
     /FIXED    = mean sal mean*sal
     /REPEATED = condition | COVTYPE(UN) SUBJECT(PersonID).
```

SPSS: /PRINT = R provides R matrix, but RCORR is not available

```
display as result "STATA Empty Multivariate Model:"
display as result "RT Mean Differences for Meaning by Salience"
display as result "Unstructured R Matrix"
mixed logrt ib(last).mean##ib(last).sal, ///
         || personid: , noconstant variance reml ///
         residuals(unstructured,t(condition)),
      estat ic, n(156),
      estat wcorrelation, covariance,
      estat wcorrelation,
      estimates store UN
```

STATA: estat ic provides AIC and BIC, where n( ) provides sample size (# persons) to be used in BIC
estat wcorrelation, covariance → R matrix
estat wcorrelation → RCORR matrix

```
TITLE1 "SAS Empty Multivariate Model: RT Mean Differences for Meaning by Salience";
TITLE2 "Unstructured R Matrix";
PROC MIXED DATA=work.Example5 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID condition mean sal;
       MODEL logrt = mean|sal@2 / SOLUTION DDFM=Satterthwaite;
       REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
RUN; TITLE1; TITLE2;
```

SAS: R and RCORR to show in output

SAS Output from Unstructured R Matrix model:

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	**0.1366**	0.1296	0.1205	0.1254
2	0.1296	**0.2369**	0.1676	0.1652
3	0.1205	0.1676	**0.2291**	0.1673
4	0.1254	0.1652	0.1673	**0.2059**

This **R matrix** holds the variances and covariances across conditions. Given complete data, it will exactly match those in original data (although complete data is not required).

Do the variances appear to differ across conditions?

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.7207	0.6814	0.7479
2	0.7207	1.0000	0.7194	0.7481
3	0.6814	0.7194	1.0000	0.7705
4	0.7479	0.7481	0.7705	1.0000

This **RCORR matrix** holds the correlations across conditions. Given complete data, it will exactly match those in the original data (although complete data is not required).

Do the correlations appear to differ across conditions?

Fit Statistics

-2 Res Log Likelihood	336.6
AIC (smaller is better)	356.6
AICC (smaller is better)	356.9
BIC (smaller is better)	387.1

This is the sum of the individual log-likelihoods multiplied by −2. It is the best possible fit for the model for the variance.

Now let's see if we could have used a simpler model: Compound Symmetry, in which all variances are predicted to be equal and all covariances are predicted to be equal, too ("Univariate" ANOVA):

```
ECHO 'SPSS Empty Multivariate Model: RT Mean Differences for Meaning by Salience;'.
ECHO 'Compound Symmetry R Matrix'.
MIXED logrt BY PersonID condition mean sal
     /METHOD   = REML
     /PRINT    = SOLUTION TESTCOV R
     /FIXED    = mean sal mean*sal
     /REPEATED = condition | COVTYPE(CS) SUBJECT(PersonID).

display as result "STATA Empty Multivariate Model:"
display as result "RT Mean Differences for Meaning by Salience"
display as result "Compound Symmetry R Matrix"
mixed logrt ib(last).mean##ib(last).sal, ///
         || personid: , noconstant variance reml ///
         residuals(exchangeable,t(condition)),
      estat ic, n(156),
      estat wcorrelation, covariance,
      estat wcorrelation,
      estimates store CS
      lrtest UN CS

TITLE1 "SAS Empty Multivariate Model: RT Mean Differences for Meaning by Salience";
TITLE2 "Compound Symmetry R Matrix";
PROC MIXED DATA=work.Example5 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID condition mean sal;
       MODEL logrt = mean|sal@2 / SOLUTION DDFM=Satterthwaite;
       REPEATED condition / R RCORR TYPE=CS SUBJECT=PersonID;
RUN; TITLE1; TITLE2;
```

SAS Output from Compound Symmetry R Matrix model:

Estimated R Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	0.2021	0.1460	0.1460	0.1460
2	0.1460	0.2021	0.1460	0.1460
3	0.1460	0.1460	0.2021	0.1460
4	0.1460	0.1460	0.1460	0.2021

This **R matrix** now predicts the residual variance to be 0.2021 regardless of condition. Part of it (0.1460) is due to mean RT differences across persons, and the rest (0.2021 – 0.1460 = 0.056) is from within-condition residual variation.

Estimated R Correlation Matrix for PersonID 1

Row	Col1	Col2	Col3	Col4
1	1.0000	0.7221	0.7221	0.7221
2	0.7221	1.0000	0.7221	0.7221
3	0.7221	0.7221	1.0000	0.7221
4	0.7221	0.7221	0.7221	1.0000

This **RCORR matrix** now predicts the residual correlation to be 0.7221 regardless of condition.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
CS	PersonID	0.1460	0.01820	8.02	<.0001
Residual		0.05617	0.003684	15.25	<.0001

This table gives the separately estimated parameters that create the R matrix pattern. Do NOT use these *p*-values!

Fit Statistics

-2 Res Log Likelihood	371.6
AIC (smaller is better)	375.6
AICC (smaller is better)	375.6
BIC (smaller is better)	381.7

Does this CS model with only 2 parameters fit worse than the UN model with 10 parameters (1 for each possible variance and covariance; −2LL = 336.6)?

−2ΔLL (8) = 371.6 – 336.6 = 35, $p < .001$, so yes, CS fits worse (UN fits better)

Now let's examine the main and interactive effects of age group and age in the older group on RT using an unstructured R matrix for the variance and covariance across the meaning*salience conditions. Note that interactions of age group by years over 65 are NOT included (and are not logically possible)!

$$LogRT_{ci} = \beta_0 + \beta_1 Mean_{ci} + \beta_2 Sal_{ci} + \beta_3 Mean_{ci} * Sal_{ci} + \beta_4 Old_i + \beta_5 Mean_{ci} * Old_i + \beta_6 Sal_{ci} * Old_i + \beta_7 Mean_{ci} * Sal_{ci} * Old_i + \beta_8 Yrs65_{ci} + \beta_9 Mean_{ci} * Yrs65_{ci} + \beta_{10} Sal_{ci} * Yrs65_{ci} + \beta_{11} Mean_{ci} * Sal_{ci} * Yrs65_{ci} + e_{ci}$$

```
ECHO 'SPSS Conditional Multivariate Model: Add Age Group and Years over 65;'.
ECHO 'Unstructured R Matrix'.
MIXED logrt BY PersonID condition mean sal WITH old yrs65
     /METHOD   = REML
     /PRINT    = SOLUTION TESTCOV R
     /FIXED    = mean sal mean*sal old old*mean old*sal old*mean*sal
                 yrs65 yrs65*mean yrs65*sal yrs65*mean*sal
     /REPEATED = condition | COVTYPE(UN) SUBJECT(PersonID).
```

SPSS: BY = categorical, WITH = continuous
No fixed effect interaction shortcuts ☹

```
display as result "STATA Conditional Multivariate Model:"
display as result "Add Age Group and Years over 65"
display as result "Unstructured R Matrix"
mixed logrt ib(last).mean##ib(last).sal##old     ///
            ib(last).mean##ib(last).sal##yrs65,  ///
         || personid: , noconstant variance reml ///
         residuals(unstructured,t(condition)),
      estat ic, n(156),
      estat wcorrelation, covariance,
      estat wcorrelation,
```

STATA: i. = categorical, c. = continuous
estimates all possible interaction and lower-order main effects

SAS: CLASS = categorical (default is continuous)
| estimates all interaction and lower-order main effects up to order specified using @

```
TITLE1 "SAS Conditional Multivariate Model: Add Age Group and Years over 65";
TITLE2 "Unstructured R Matrix";
PROC MIXED DATA=work.Example5 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
       CLASS PersonID condition mean sal;
       MODEL logRT = mean|sal|old@3 mean|sal|yrs65@3 / SOLUTION DDFM=Satterthwaite;
       REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
RUN; TITLE1; TITLE2;
```

Relevant SAS Output, treating meaning and salience as "categorical" but old and yrs65 as "continuous" so that it will not marginalize across age in estimating marginal effects of meaning and salience:

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
mean	1	153	48.81	48.81	<.0001	<.0001
sal	1	153	236.14	236.14	<.0001	<.0001
mean*sal	1	153	13.13	13.13	0.0003	0.0004
old	1	153	137.89	137.89	<.0001	<.0001
old*mean	1	153	1.60	1.60	0.2059	0.2079
old*sal	1	153	0.06	0.06	0.8046	0.8049
old*mean*sal	1	153	0.51	0.51	0.4764	0.4774
yrs65	1	153	16.03	16.03	<.0001	<.0001
yrs65*mean	1	153	0.02	0.02	0.8970	0.8971
yrs65*sal	1	153	3.01	3.01	0.0828	0.0848
yrs65*mean*sal	1	153	1.43	1.43	0.2324	0.2342

Because old and yrs65 are continuous, these are the effects for younger adults.

These are how the meaning and salience effects DIFFER in the older adult group (conditional at age 65 years).

These are how the meaning and salience effects DIFFER per additional year of age in the older adult group.

Based on these results, it appears we can remove some fixed effects, starting with yrs65*mean*sal. The two-way interactions of yrs65*mean and yrs65*sal were still not significant, so those were removed, leaving only the significant main effect of yrs65.

Here is the reduced model (in which the highest-order interaction is significant):

$$LogRT_{ci} = \beta_0 + \beta_1 Mean_{ci} + \beta_2 Sal_{ci} + \beta_3 Mean_{ci} * Sal_{ci} + \beta_4 Old_i + \beta_5 Mean_{ci} * Old_i + \beta_6 Sal_{ci} * Old_i + \beta_7 Mean_{ci} * Sal_{ci} * Old_i + \beta_8 Yrs65_{ci} + e_{ci}$$

```
ECHO 'SPSS Reduced Conditional Multivariate Model: Years over 65 as Main Effect;'.
ECHO 'Unstructured R Matrix'.
MIXED logrt BY PersonID condition mean sal WITH old yrs65
     /METHOD   = REML
     /PRINT    = SOLUTION TESTCOV R
     /FIXED    = mean sal mean*sal old old*mean old*sal old*mean*sal yrs65
     /REPEATED = condition | COVTYPE(UN) SUBJECT(PersonID)
     /EMMEANS  = TABLES(mean*sal) COMPARE(mean) WITH(old=0 yrs65=0)
     /EMMEANS  = TABLES(mean*sal) COMPARE(mean) WITH(old=1 yrs65=0)
     /EMMEANS  = TABLES(mean*sal) COMPARE(mean) WITH(old=1 yrs65=10)
     /EMMEANS  = TABLES(mean*sal) COMPARE(sal)  WITH(old=0 yrs65=0)
     /EMMEANS  = TABLES(mean*sal) COMPARE(sal)  WITH(old=1 yrs65=0)
     /EMMEANS  = TABLES(mean*sal) COMPARE(sal)  WITH(old=1 yrs65=10)
     /TEST= "Old: Low Mean,  Low Sal"  old 1 mean*old 1 0 sal*old 1 0 mean*sal*old 1 0 0 0
     /TEST= "Old: Low Mean,  High Sal" old 1 mean*old 1 0 sal*old 0 1 mean*sal*old 0 1 0 0
     /TEST= "Old: High Mean, Low Sal"  old 1 mean*old 0 1 sal*old 1 0 mean*sal*old 0 0 1 0
     /TEST= "Old: High Mean, High Sal" old 1 mean*old 0 1 sal*old 0 1 mean*sal*old 0 0 0 1.
```

SPSS: EMMEANS gives conditional means, TEST gets slopes for age group per condition

```
display as result "STATA Reduced Conditional Multivariate Model:"
display as result "Years over 65 as Main Effect"
display as result "Unstructured R Matrix"
mixed logrt ib(last).mean##ib(last).sal##old yrs65, ///
         || personid: , noconstant variance reml ///
         residuals(unstructured,t(condition)),
      estat ic, n(156),
      estat wcorrelation, covariance,
      estat wcorrelation,
```

```
margins ib(last).mean#ib(last).sal, at(c.old=0 c.yrs65=0)
margins ib(last).mean#ib(last).sal, at(c.old=1 c.yrs65=0)
margins ib(last).mean#ib(last).sal, at(c.old=1 c.yrs65=10)
margins ib(last).mean@ib(last).sal, at(c.old=0 c.yrs65=0)
margins ib(last).mean@ib(last).sal, at(c.old=1 c.yrs65=0)
margins ib(last).mean@ib(last).sal, at(c.old=1 c.yrs65=10)
margins ib(last).sal@ib(last).mean, at(c.old=0 c.yrs65=0)
margins ib(last).sal@ib(last).mean, at(c.old=1 c.yrs65=0)
margins ib(last).sal@ib(last).mean, at(c.old=1 c.yrs65=10)
lincom c.old*1 + i1.mean#c.old*1 + i1.sal#c.old*1 + i1.mean#i1.sal#c.old*1
lincom c.old*1 + i1.mean#c.old*1 + i2.sal#c.old*1 + i1.mean#i2.sal#c.old*1
lincom c.old*1 + i2.mean#c.old*1 + i1.sal#c.old*1 + i2.mean#i1.sal#c.old*1
lincom c.old*1 + i2.mean#c.old*1 + i2.sal#c.old*1 + i2.mean#i2.sal#c.old*1
```

STATA: margins gets conditional means, lincom gets slopes for age group per condition

```
TITLE1 "SAS Reduced Conditional Multivariate Model: Years over 65 as Main Effect";
TITLE2 "Unstructured R Matrix";
PROC MIXED DATA=work.Example5 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
      CLASS PersonID condition mean sal;
      MODEL logRT = mean|sal|old@3 yrs65 / SOLUTION DDFM=Satterthwaite;
      REPEATED condition / R RCORR TYPE=UN SUBJECT=PersonID;
* Getting condition means and simple effect tests at different ages;
      LSMEANS mean*sal / AT (old yrs65)=(0  0) SLICE=mean SLICE=sal; * For YA;
      LSMEANS mean*sal / AT (old yrs65)=(1  0) SLICE=mean SLICE=sal; * For age 65;
      LSMEANS mean*sal / AT (old yrs65)=(1 10) SLICE=mean SLICE=sal; * For age 75;
* Getting age group differences per condition -- need all terms with old slope in them;
ESTIMATE "Old: Low Mean,  Low Sal"  old 1 mean*old 1 0 sal*old 1 0 mean*sal*old 1 0 0 0;
ESTIMATE "Old: Low Mean,  High Sal" old 1 mean*old 1 0 sal*old 0 1 mean*sal*old 0 1 0 0;
ESTIMATE "Old: High Mean, Low Sal"  old 1 mean*old 0 1 sal*old 1 0 mean*sal*old 0 0 1 0;
ESTIMATE "Old: High Mean, High Sal" old 1 mean*old 0 1 sal*old 0 1 mean*sal*old 0 0 0 1;
RUN; TITLE1; TITLE2;
```

Relevant SAS Output:

Rows with 0's and dots are redundant effects not estimated for the reference group (YA in the high–high condition)

Solution for Fixed Effects

Effect	Meaning (1=Low, 2=High)	Salience (1=Low, 2=High)	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept			1.6768	0.02533	154	66.19	<.0001
mean	1Low		0.08866	0.03410	154	2.60	0.0102
mean	2High		0	.	.	.	.
sal		1Low	0.2566	0.03220	154	7.97	<.0001
sal		2High	0	.	.	.	.
mean*sal	1Low	1Low	0.1706	0.04714	154	3.62	0.0004
mean*sal	1Low	2High	0	.	.	.	.
mean*sal	2High	1Low	0	.	.	.	.
mean*sal	2High	2High	0	.	.	.	.
old			0.6181	0.05847	221	10.57	<.0001
old*mean	1Low		0.03675	0.05545	154	0.66	0.5085
old*mean	2High		0	.	.	.	.
old*sal		1Low	0.04123	0.05236	154	0.79	0.4322
old*sal		2High	0	.	.	.	.
old*mean*sal	1Low	1Low	-0.2510	0.07665	154	-3.27	0.0013
old*mean*sal	1Low	2High	0	.	.	.	.
old*mean*sal	2High	1Low	0	.	.	.	.
old*mean*sal	2High	2High	0	.	.	.	.

How years of age adjusts the intercept in older adults (is same for all conditions)

yrs65			0.01425	0.003820	153	3.73	0.0003

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
mean	1	154	49.12	49.12	<.0001	<.0001
sal	1	154	233.10	233.10	<.0001	<.0001
mean*sal	1	154	13.09	13.09	0.0003	0.0004
old	1	155	147.11	147.11	<.0001	<.0001
old*mean	1	154	4.84	4.84	0.0279	0.0294
old*sal	1	154	5.36	5.36	0.0206	0.0220
old*mean*sal	1	154	10.72	10.72	0.0011	0.0013

Because old and yrs65 are continuous, these are the effects for younger adults.

These are how the meaning and salience effects DIFFER in the older adult group.

Estimates

These are the simple slopes for how the YA and OA groups differ in mean RT for each condition.

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Old: Low Mean, Low Sal	0.4451	0.05595	207	7.96	<.0001
Old: Low Mean, High Sal	0.6549	0.06188	231	10.58	<.0001
Old: High Mean, Low Sal	0.6594	0.06042	228	10.91	<.0001
Old: High Mean, High Sal	0.6181	0.05847	221	10.57	<.0001

Least Squares Means

These are the conditional means per condition for each level of age requested (YA, 65, and 75).

Effect	Meaning (1=Low, 2=High)	Salience (1=Low, 2=High)	old	yrs65	Estimate	Standard Error	DF	t Value	Pr > |t|
mean*sal	1Low	1Low	0.00	0.00	2.1926	0.02308	153	95.01	<.0001
mean*sal	1Low	2High	0.00	0.00	1.7654	0.02823	153	62.54	<.0001
mean*sal	2High	1Low	0.00	0.00	1.9334	0.02701	154	71.59	<.0001
mean*sal	2High	2High	0.00	0.00	1.6768	0.02533	154	66.19	<.0001
mean*sal	1Low	1Low	1.00	0.00	2.6377	0.05097	200	51.75	<.0001
mean*sal	1Low	2High	1.00	0.00	2.4203	0.05506	231	43.95	<.0001
mean*sal	2High	1Low	1.00	0.00	2.5927	0.05405	225	47.97	<.0001
mean*sal	2High	2High	1.00	0.00	2.2949	0.05270	215	43.55	<.0001
mean*sal	1Low	1Low	1.00	10.00	2.7802	0.02977	155	93.38	<.0001
mean*sal	1Low	2High	1.00	10.00	2.5628	0.03634	154	70.51	<.0001
mean*sal	2High	1Low	1.00	10.00	2.7352	0.03479	156	78.63	<.0001
mean*sal	2High	2High	1.00	10.00	2.4374	0.03265	156	74.66	<.0001

Tests of Effect Slices

These are the simple effects of condition within each level of age requested. Note they are same within the 65- and 75-year-olds….

Effect	Meaning (1=Low, 2=High)	Salience (1=Low, 2=High)	old	yrs65	Num DF	Den DF	F Value	Pr > F
mean*sal	1Low		0.00	0.00	1	154	169.39	<.0001
mean*sal	2High		0.00	0.00	1	154	63.51	<.0001
mean*sal		1Low	0.00	0.00	1	154	56.94	<.0001
mean*sal		2High	0.00	0.00	1	154	6.76	0.0102
mean*sal	1Low		1.00	0.00	1	154	26.69	<.0001
mean*sal	2High		1.00	0.00	1	154	52.04	<.0001
mean*sal		1Low	1.00	0.00	1	154	1.04	0.3088
mean*sal		2High	1.00	0.00	1	154	8.23	0.0047
mean*sal	1Low		1.00	10.00	1	154	26.69	<.0001
mean*sal	2High		1.00	10.00	1	154	52.04	<.0001
mean*sal		1Low	1.00	10.00	1	154	1.04	0.3088
mean*sal		2High	1.00	10.00	1	154	8.23	0.0047